# $Id: PKGBUILD 137772 2011-09-10 18:35:00Z eric $
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Jochem Kossen <j.kossen@home.nl>

pkgname=id3
pkgver=0.78
pkgrel=3
pkgdesc="Utility to edit id3v1 and id3v2 tags"
arch=('i686' 'x86_64')
url="http://freshmeat.net/projects/id3"
license=('BSD')
depends=('gcc-libs')
options=('!makeflags')
source=(http://linux-bsd-unix.strefa.pl/${pkgname}-${pkgver}.tar.gz
        id3-0.78-gcc44.patch)
md5sums=('bbc2a5d8022006330e6bb51034fafb43'
         '51370be8d2624efee3683efab608a7f1')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -Np1 -i ${srcdir}/id3-0.78-gcc44.patch
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make prefix="${pkgdir}/usr" mandir="${pkgdir}/usr/share/man" install

  #install license
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}

